Is an Equilibrium Unique? Why or why not?

TheoryGuru applications

Background

To query uniqueness,
(i) use the equilibrium conditions to define an equilibrium v_1 (v_1 could be a list, aka vector),
(ii) use the same conditions to define v_2 as an equilibrium,
(iii) ask TheoryGuru if v_1 is the same as v_2 (for scalars, this is simply $v_1 == v_2$; see GlobalOptimumofConcaveFunction.nb (pdf here) for vectors).

If True, then the equilibrium is unique.

Alternatively, assume that v_1 is not the same as v_2. That will be a contradictory assumption if and only if the equilibrium is unique.

If TheoryGuru's answer to $v_1 == v_2$ is "True for some, False for Others," then TheorySufficient and TheoryExtra can be used to yield conditions that guarantee a unique equilibrium.

Setup

```
Get@"http://economicreasoning.com"
```

Proof & Logic Tools 6.1

Type ERCommands for a list of commands in the package.

Introduction to Automated Economic Reasoning		
Tutorials:	Entering calculus	General Mathematica tips
Get started	Load extras	Browse examples

```
SetOptions[TheoryGuru, keepall → True];
```

Load other tools by clicking on extras and/or evaluating below

```
If[Not@MemberQ[$ContextPath, "OtherTools`"],
 Get["http://othertools.economicreasoning.com"]]
```

Definitions for supply-demand example

```
DemandSlopesDown = (d[p2] - d[p1]) (p2 - p1) < 0 ∨ d[p2] - d[p1] == 0 == p2 - p1;

SupplySlopesUp = (s[p2] - s[p1]) (p2 - p1) > 0 ∨ s[p2] - s[p1] == 0 == p2 - p1;

Equilibrium[p_] := d[p] == s[p]

UniqueEquilibrium = p1 == p2;

MultipleEquilibria = Not[UniqueEquilibrium];
```

These are unnecessary but keep the examples realistic

```
signconditions = {p1 > 0, p2 > 0, d[p1] > 0, d[p2] > 0, s[p1] > 0, s[p2] > 0};
```

Text interpretations of key inequalities

```
rtext = {Simplify@LogicalExpand[DemandSlopesDown ∧ SupplySlopesUp] →
     "\"Demand slopes down and supply slopes up.\"",
   Simplify@LogicalExpand[Not@DemandSlopesDown ∨ Not@SupplySlopesUp] →
     "\"Demand slopes up or supply slopes down\" (sic)",
   MultipleEquilibria → "Multiple equilibria.",
   UniqueEquilibrium → "Unique equilibrium"};
```

Supply-Demand Results

Demand slopes down by itself is consistent with MultipleEquilibrium

```
TheoryGuru[{Equilibrium@p1, Equilibrium@p2, DemandSlopesDown,
   signconditions},
 UniqueEquilibrium]

True for some, False for others

TheoryEmpty[{Equilibrium@p1, Equilibrium@p2, DemandSlopesDown,
   signconditions, MultipleEquilibria}]

False
```

```
TheoryInstance[{Equilibrium@p₁, Equilibrium@p₂, DemandSlopesDown,
  signconditions},
 MultipleEquilibria]
```

d[p_1]	d[p_2]	s[p_1]	s[p_2]	p_1	p_2
2	1	2	1	$\frac{1}{2}$	1

But MultipleEquilibrium requires either demand sloping up or supply sloping down (sic).

```
TheoryGuru[{Equilibrium@p₁, Equilibrium@p₂, DemandSlopesDown, SupplySlopesUp},
 UniqueEquilibrium]
```

True

```
Simplify@TheoryOverlap[{Equilibrium@p₁, Equilibrium@p₂},
   DemandSlopesDown ∧ SupplySlopesUp,
   UniqueEquilibrium] /. rtext
```

Unique equilibrium	is necessary but not sufficient for	"Demand slopes down and supply slopes up."

```
Simplify@TheoryOverlap[{Equilibrium@p₁, Equilibrium@p₂},
   Not@DemandSlopesDown ∨ Not@SupplySlopesUp,
   MultipleEquilibria] /. rtext
```

"Demand slopes up or supply slopes down" (sic)	is necessary but not sufficient for	Multiple equilibria.

Use TheorySufficient to discover a supply-curve restriction that is sufficient to guarantee a unique equilibrium

```
TheorySufficient[{Equilibrium@p₁, Equilibrium@p₂, DemandSlopesDown,
  signconditions},
 UniqueEquilibrium,
 {s[p₁], s[p₂], p₁, p₂}]
```

$(s[p_1] - s[p_2])\,(p_1 - p_2) \geq 0$

If one curve is concave and the other is convex, then there can be two equilibria but not three

```
ThreeEquilibria = MultipleEquilibria ∧ p₃ ≠ p₁ ∧ p₃ ≠ p₂
```

$p_1 \neq p_2$ && $p_3 \neq p_1$ && $p_3 \neq p_2$

```
TheoryGuru[{Equilibrium /@ {p1, p2, p3},
  GloballyConcaveFunction[d, {p1, p2, p3}, strictly → True],
  GloballyConvexFunction[s, {p1, p2, p3}, strictly → True],
  MultipleEquilibria},

 Not[ThreeEquilibria]]
```

```
True
```

```
TheoryGuru[{Equilibrium /@ {p1, p2, p3},
  GloballyConcaveFunction[s, {p1, p2, p3}, strictly → True],
  GloballyConvexFunction[d, {p1, p2, p3}, strictly → True],
  MultipleEquilibria},

 Not[ThreeEquilibria]]
```

```
True
```

```
TheoryGuru[{Equilibrium /@ {p1, p2},
  GloballyConcaveFunction[d, {p1, p2, p3}, strictly → True],
  GloballyConvexFunction[s, {p1, p2, p3}, strictly → True],
  MultipleEquilibria},

 p3 == p1 ∨ p3 == p2 ∨ Not@Equilibrium@p3]
```

```
True
```

```
TheoryGuru[{Equilibrium /@ {p1, p2},
  GloballyConcaveFunction[s, {p1, p2, p3}, strictly → True],
  GloballyConvexFunction[d, {p1, p2, p3}, strictly → True],
  MultipleEquilibria},

 p3 == p1 ∨ p3 == p2 ∨ Not@Equilibrium@p3]
```

```
True
```

Look “under the hood” at the Tarski formula automatically assembled by TheoryGuru

Variable interpretations

Laffer Curve Results

See also LafferCurveSurprises.nb (pdf here)

Setup

```
basicassumptions = {0 < τL ≤ τH < 1, wL > 0, wH > 0, nL > 0, nH > 0, cL > 0, cH > 0,
   SameSign[cL - cH, nL - nH], SameSign[wH - wL, nL - nH], (nL - nH) wL ≤ cL - cH ≤ (nL - nH) wH,
   uL > uLb ∨ nL == nH, SameSign[uLb - uH, cLb - cH], cL - cLb == (nL - nH) (1 - τL) wL,
   uH > uHb ∨ nL == nH, SameSign[uHb - uL, cHb - cL], cH - cHb == (nH - nL) (1 - τH) wH};
```

```
mrsassumptions = {(nL == nH ∧ cL == cH) ⇒ (mL == mH)
   (* m is the marginal rate of substitution, which is a function of c and n *),
   mL == (1 - τL) wL, mH == (1 - τH) wH};
```

```
equalrevenue = τL nL wL == τH nH wH;
```

```
ldemandiselastic = (wL nL - wH nH) (nL - nH) ≥ 0;
```

```
bothnormalgoods = SameSign[mL - mH, nL - nH];
```

There is only one policy that can support a given amount of labor as an equilibrium

```
TheoryGuru[{basicassumptions[[3 ;; 10]], mrsassumptions,
   nL == nH},
  τL == τH ∧ cL == cH ∧ equalrevenue]
```

True

```
TheoryOverlap[{basicassumptions[[3 ;; 10]], mrsassumptions},
  nL == nH,
  τL == τH ∧ cL == cH ∧ equalrevenue]
```

$\{n_L = n_H,\ c_L = c_H \wedge \tau_L = \tau_H \wedge n_L\, \tau_L\, w_L = n_H\, \tau_H\, w_H\}$ are equivalent

```
TheoryOverlap[{basicassumptions[[3 ;; 10]], mrsassumptions},
  cL == cH,
  τL == τH ∧ nL == nH ∧ equalrevenue]
```

$\{c_L = c_H,\ n_L = n_H \wedge \tau_L = \tau_H \wedge n_L\, \tau_L\, w_L = n_H\, \tau_H\, w_H\}$ are equivalent

More than one equilibrium allocation is consistent with the same revenue

```
TheoryGuru[{basicassumptions[[1 ;; 10]], τL < τH,
   equalrevenue},
  nL ≠ nH ∧ cL ≠ cH]
```

True

The low tax may produce less labor, but this can be ruled out by assuming that both goods are normal

```
TheoryOverlap[{basicassumptions[[1 ;; 10]], mrsassumptions, τL < τH, equalrevenue},
 nL > nH,
 bothnormalgoods]
```

$n_L > n_H$ is necessary but not sufficient for $(m_L - m_H = 0 \wedge n_L - n_H = 0) \vee (m_L - m_H)(n_L - n_H) > 0$

... or by assuming that labor demand is elastic

```
TheoryOverlap[{basicassumptions[[1 ;; 10]], τL < τH, equalrevenue},
 nL > nH,
 ldemandiselastic]
```

$\{n_L > n_H, (n_L - n_H)(n_L w_L - n_H w_H) \geq 0\}$ are equivalent

```
TheoryGuru[{basicassumptions[[1 ;; 10]], mrsassumptions,
  τL < τH, equalrevenue, Not@bothnormalgoods, nL < nH},
 Not@ldemandiselastic]
```

True

```
TheoryGuru[
 {basicassumptions[[1 ;; 10]], mrsassumptions, τL < τH, equalrevenue, nL < nH},
 Not@bothnormalgoods ∧ Not@ldemandiselastic]
```

True

More than one equilibrium revenue amount is consistent with the same tax rate, but this can be ruled out by assuming that both goods are normal

```
TheoryGuru[{basicassumptions[[1 ;; 10]], mrsassumptions, τL == τH, Not@equalrevenue},
 Not@bothnormalgoods]
```

True

i.e., when both goods are normal, there is a unique mapping from tax rate to revenue

```
TheoryGuru[{basicassumptions[[1 ;; 10]], mrsassumptions, τL == τH},
 equalrevenue ∨ Not@bothnormalgoods]
```

True

Variable interpretations